
The problem of detecting feature interactions in telephone systems design is
addressed. The method proposed involves specification of the features in LOTOS,
and uses an analysis technique called backward reasoning. This is  is imple-
mented in LOTOS by a combination of  backward and forward execution. A tool to
help carry out backward execution is presented. A detailed example of the use of
the technique is given, involving the three-way-calling and call-waiting features.

0. OVERVIEW

Feature interactions have been categorized according to the kinds of features,
the number of users and the number of network elements [CGLN93]. The causes
of feature interactions are equally varied. They mainly revolve around problems
of ambiguity and errors in logic intrinsic to distributed systems. It is reasonable
to think that different causes require different detection methods. This paper
concentrates on the problem of detecting ambiguous actions, which are seen as
symptoms of feature interactions. Features are specified in LOTOS, and the spec-
ification is analyzed by using backward reasoning [DB78] [Hol85] [Lin 90].

1. SPECIFICATION OF FEATURES USING THE STATUS ORIENTED
STYLE

There are several published LOTOS specifications of telephones with features
[BL93][Najm93]. They often use the constraint-oriented specification style where
abstract data types play an important control role. The status-oriented style
[SL93], a variation of the resource-oriented style, makes greater use of the princi-
ple of synchronization on discrete values, which usually represent signals and
phone numbers. When complex control structures need to be specified, the logic is

1. Appeared in: S. Vuong and S. Chanson (eds). Protocol Specification, Testing, and Verification, XIV
(Chapman and Hall, 1995), 71-86.

encoded into separate processes that work as constraints on a local basis, corre-
sponding to what one would use as implementation structure. In the constraint-
oriented style, all processes collaborate together to form the global behavior. Thus
the behavior expressions representing features are mixed together and it may
become difficult to follow the life cycle of a feature. The status-oriented style is
closer to the concept of components communicating via interfaces, a concept that
is central in ODP.

 The specification we have used describes three cases of feature interactions
that are illustrations of the classification of feature interactions found in
[CGLN93].

The Single User Multiple Element (SUME) class is represented by the case of
the interaction caused when a user presses the pound key to change her personal
options in voice mail when accessing it through a calling card call (action which
also involves the use of the pound key).

The Single User Single Element (SUSE) class is represented by the case of the
interaction caused when a user attempting to establish a three way call presses
the flash_hook key while being called by another user and after having sub-
scribed to call waiting. The flash-hook key becomes ambiguous because it can also
be used to answer the waiting call.

The Multiple User Single Element (MUSE) class is represented by the case of
the interaction caused when a user has a number in her originate screen list that
is the number to which another number this user is calling is forwarded to.

The two first cases are essentially cases of detections of ambiguities while the
third case is a violation of intentions.

While developing the specification we have been faced with one interesting
reality: for specification of features in LOTOS one can observe the same phenom-
enon as for implementation of features in software: features are scattered
throughout the code. However while code can be tested only by running various
test scenarios, LOTOS (due to its algebraic nature) can be manipulated to verify
properties. Also the very activity of specifying features using LOTOS forces the
designer to think more and thus detect feature interactions just by trying to spec-
ify them. One difference is that the cost of experimenting with specifications is
considerably less than the cost of experimenting with real software or hardware.

The general structure of the specification is as follows:

 (
      phone[u, n](num1, { three_way_calling, call_waiting,originate_screening }, no_active_service , num1, {
num4 } )
    |||
     phone[u, n](num2, { call_forward }, no_active_service ,num4, no_screen_list)
    |||
     voice_mail[u, n](num3)
    |||
     phone[u, n](num4,no_services, no_active_service,num4, no_screen_list)
    |||
     phone[u, n](num5,no_services, no_active_service,num5, no_screen_list)
 )

|[n]|

 network[n,a]

The formal parameter list of each instance of phone determines the activation
of the various services presented here. Each phone is defined as a choice of call
initiator and call responder. The process network allows to manage simultaneous
connections and is of the usual following form:

processnetwork[n,a]:noexit:=
    n ? CALLER:phone_number ! tone  ;    ( connection_handler[n,a](CALLER)    |||    network[n,a]  )
endproc

The features are inserted wherever they can be invoked within the basic
POTS behaviors. For example, in the call_initiator_phone behavior, the call-wait-
ing and three-way-calling features are inserted after the connect event in process
complete_connection as an interleaving between instances of processes user_-
events and features.

processcall_initator_phone[...](...) := off_hook ; tone ; dial ; complete_connection[...](...)
where
processcomplete_connection[u,n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=

       n ! NUMBER ! connect
       ;     (  user_events[u,n](NUMBER,SUBS_SERV) |||  features[u,n](NUMBER,SUBS_SERV)     )
endproc

The process features is itself an interleaving of the two features.
The structuring of feature interactions proposed in [CGLN93] has been  useful

to explore the problem using LOTOS. It enabled us to isolate problems and to
concentrate on them. This idea was further developed in the concept of backward
execution of a specification. By using this method, we can produce rapidly a sce-
nario where an interaction can be found under the guidance of the designer. The
method consists in decomposing problems into sub-problems to isolate scenarios
that are likely to generate interactions.

2. DETECTING AMBIGUITIES BY BACKWARD REASONING

2.1 Ambiguous actions
We say that an observable action in a LOTOS specification is ambiguous if in

the behavior tree of the specification there is a branching point where the action
is the first observable one in at least two branches. Ambiguity represents nonde-
terministic behavior of the ‘black box’ being specified: it represents the situation
where the user presses a button, and one of two different effects can follow.
Ambiguous actions can be offered in several cases: most obvious are the cases
where two behaviors are specified using the choice [] or disable [> operator. These
are easy to detect, thus they are of limited interest. The most important case is
when two behaviors are  interleaved and both contain the same action, leading to
the possibility that one of the many combinations of interleavings produces a sit-



4 March 1, 1995 6:15 pm

uation of ambiguity.
As a trivial example, in the following behavior

a ; b ; c ; stop
 |||

  e ; b ; f ; stop

after trace <ae> we end up with

b ; c ; stop
           |||

b ; f ; stop

We can think that b here represents an ambiguous key, leading to two behav-
iors  that represent two different features. Because of the way the specification is
written, a disorderly mix of the two features results if the user presses b. One can
easily construct more complex examples involving process instantiations, etc.
where the nondeterminism is far from obvious. In [BL93] such an example is
given. We give a similar example in this paper, however much  simpler and  spec-
ified in a different style.

We concentrate on the problem of detecting cases of ambiguity, seen as symp-
toms of feature interaction. Note that in general ambiguity and nondeterminism
can involve internal actions, however these  will be ignored here for simplicity of
discussion.

2.2 Case studies
The case of call-waiting and three-way calling is relatively simple. Both

involve the action flash_hook. Since the two processes are interleaved, it is clear
that an interaction could exist.

Fig 1.  specification and resulting execution paths

processfeatures[u, n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=
      call_waiting[u,n](NUMBER,SUBS_SERV)
     |||
      three_way_calling[u,n](NUMBER,SUBS_SERV)
endproc

where

processcall_waiting[u,n](NUMBER:phone_number, SUBS_SERV:subscrib_services):noexit:=
       [ call_waiting IsIn SUBS_SERV] ->
                  n ! NUMBER ! ring
                ;u ! NUMBER ! flash_hook
                ; n ! NUMBER ! connect
               ;
                  ( user_events[u,n](NUMBER,SUBS_SERV)
                      [>
                        (     n ! discon_req
                            ;stop
                         )
                    )
endproc

processthree_way_calling[u,n](NUMBER:phone_number,
                                                       SUBS_SERV:subscrib_services):noexit:=
      [ three_way_calling IsIn SUBS_SERV] ->

u ! NUMBER ! flash_hook
                 ; u ! NUMBER ! dial ? C:phone_number
                 ; n ! NUMBER ! conreq ! C
                 ; n ! NUMBER ! connect
                ;
                  ( user_events[u,n](NUMBER,SUBS_SERV)
                       [>
                            (    n ! discon_req
                              ;stop
                            )
                  )
endproc

Note that the simplest case of call-waiting sequence has been proposed. In
reality one should be able to answer call waiting immediately after receiving a
dial tone. We found that it is difficult to specify this more general case by using
existing LOTOS operators. This is because when a subscriber is involved in a fea-
ture, the other features can sometimes be disabled, something that cannot be
expressed easily with the interleave operator. In order to deal with such cases, we
have proposed a new suspend and resume operator for LOTOS, which is described
in [CDN93].

The SUME case of ambiguity, resulting when a user tries to change the per-
sonalized options of the voice mail feature while a card connection is being used,
is more complex because the ambiguity of the pound key action is due to two pro-
cesses that are not directly interleaved. The interleaved situation is entirely con-
tained in process network_complete_connection via instances of processes

calling_card_intercept and relay_user_events where the predicate is_user_actions
could be evaluated with respect to the value pound_key. This case shows that it is
not always easy to find feature interactions directly.

processnetwork_complete_connection[n,a] (CALLER,CALLED:phone_number):noexit:=
            n ! CALLED ! ring
          ; n ! CALLED ! connect
          ; n ! CALLER ! connect
         ;
           (
             relay_user_events[n](CALLER,CALLED)
            |||
            calling_card_intercept[n,a](CALLER)
          )
endproc

where
processrelay_user_events[n] (CALLER,CALLED:phone_number):noexit:=

 n ! CALLER ? EVENT:primitives [ is_user_actions(EVENT) ]
        ; n ! CALLED ! EVENT
        ; relay_user_events[n](CALLER,CALLED)

     []
          ...
endproc

processcalling_card_intercept[n,a](CALLER:phone_number):noexit:=
n ! CALLER ! pound

       ; a ! CALLER ! play_announce_new_number
       ; n ! CALLER ? NEW_NUMBER:phone_number ! conreq
       ; network_complete_connection[n,a](CALLER,NEW_NUMBER)
endproc

processvoice_mail[v, n](NUMBER:phone_number):noexit:=
        n ! NUMBER ! ring
   ;    v ! NUMBER ! voice_mail_answer
   ;    n ! NUMBER ! connect
   ;    v ! NUMBER ! play_announce_pwd
   ;    n ! NUMBER ! pwd
   ;    v ! NUMBER ! deliver_messages
   ;    v ! NUMBER ! play_announce_star_pound
   ;
   (
        n ! NUMBER ! star
     ;  v ! Number ! good_bye
     ; stop
    []
        n ! NUMBER ! pound
     ;  v ! NUMBER ! play_announce_management
     ; stop
   )
endproc

It is the evaluation of the predicate [ is_user_actions(EVENT) ] that risks pro-
ducing an ambiguity with respect to the two actions in boldface. If the variable
EVENT captures a pound sign, the predicate will be true and thus the action will
be executable. It is process voice_mail that plays the role of the environment and
decides this case.

2.3 The method
Most LOTOS tools available today are based on systematic use of inference

rules or expansion. In both cases one runs sooner or later into state explosion.
Some new techniques such as goal oriented execution [HLS93] [BE93] and inter-
leaved expansion [QLP93] provide partial relief to this problem.

The method presented here consists in exploring a specification by focusing on
detection of ambiguities when a new feature is introduced, thus exercising only
portions of the specification that are relevant to the problem. This is achieved by
using well-known concepts and techniques, but in a particular manner. We
decompose the specification, we find local traces, we substitute them into the
overall behavior expression, and then conventional stepwise techniques are
applied. Therefore, this is an incremental method.

 The method consists of three steps:
Step 1: collect the various instances of an action found in different  features

throughout the specification.
Step 2: verify if the behavior expressions that contain these identical actions

are interleaved.
Step 3: apply backward reasoning to verify if there are cases where the

behavior expressions found in step 2 that are interleaved can produce non-
deterministic choices when these behaviors are executed under the con-
straints of their environment. This means verifying if other behaviors that
are in parallel with them can lead to such a case. In other words the fact
that two behaviors containing the same action are interleaved is a neces-
sary but not a sufficient condition.

Note that we have made the hypothesis that features are interleaved, and
that their simultaneous occurrence is pathological, which seems to be true in
most cases.

The above method is relevant to detect ambiguity of signals. There are other
cases of non-determinism that are not caused by ambiguities of signals but by
errors in logic. In LOTOS these reveal themselves as incomplete or contradictory
abstract values in guards and/or predicates. In these cases, step 1 is no longer
necessary, and the method consists in finding features that are interleaved in
step 2 and then applying the procedures of step 3.

Step 3 has been implemented in a tool, which is discussed below.

2.4 Backward reasoning in LOTOS
As mentioned, the simple fact that an action is in interleave with itself  is not suf-
ficient for two behaviors to lead to ambiguities. It is also necessary that paths
leading to the ambiguity be simultaneously possible under a given set of condi-
tions. The fact that two events that are interleaved can create non-determinism

can be proved by executing the specification to reach the first case and then,
using the resulting behavior expression, trying to execute the second case. This
can be done in several ways, for example by using already known techniques such
as forward step-by-step execution in combination with automatic generation of
symbolic trees and goal-oriented execution [HLS93][BE93].However these tech-
niques will produce state explosion to various degrees. Most of the state explosion
is the result of interleaved behavior expressions. The multiple combinations of
solutions all lead to the same desired state where our two ambiguous actions are
possible. Backward reasoning allows to find the two paths necessary to reach our
ambiguous case without having to worry with the problems created by interleav-
ing.

Our method of ‘backward reasoning” appears to be effective in many cases. In
a step-by-step, interactive way, LOTOS specifications can be executed backward.
This is due to the fact that the language has no side effects, with the exception of
those that determine the order of actions in traces. One can start from a selected
behavior expression and travel in reverse direction through action prefix, choice,
disable, and enable operators.Whenever a new variable is encountered, a value
for it is provided.This value may fail because of predicates encountered later;
however the same thing can also occur in forward execution.When interleaved
behavior expressions are encountered, they are attached to the behavior tree
developed so far, since they could be needed for synchronization later.When a syn-
chronization operator is encountered, the trace developed so far needs to be vali-
dated, so it needs to be executed forward with respect to the synchronized
behavior.This may require synchronization with some of the interleaved behavior
expressions attached earlier. There can be several such behavior expressions,
both on the trace side and on the synchronized behavior side. Note also that this
synchronization may depend on appropriate values having been entered earlier.
Goal-oriented execution can be used to facilitate the process. When a process dec-
laration is encountered, we must search all the instantiations of that process in
the behavior expression, choose one of them, and decide on parameters passed

Fig 2. stepwise execution of features

(this is the only case of choice in backward execution, which otherwise is deter-
ministic). If we can travel all the way to the root of the main behavior declaration,
we have found an executable path. Just as for regular forward execution, exe-
cuted actions are removed from the behavior expression. Obviously, as in forward
step-by-step execution, success of the procedure depends on the insight of the
user who must select meaningful paths and values to build an appropriate sce-
nario.

The concept of backward execution of protocols and service specifications has
already been studied in [DB78] [Hol85], and more recently, by using logic pro-
gramming concepts related to ours, in [Lin90].

From the above, it should be clear that backward reasoning uses both back-
ward execution and conventional forward execution. While backward execution
requires specially designed tools, forward execution can be done with  existing
interpreters and expanders, although in our experience goal-oriented execution is
a necessity.

2.5 Applying the method
In this section, we present the application of the method to the SUSE case of

ambiguity of the flash_hook in the three-way calling and call-waiting case. First
we assume that we have a fully functioning specification of the three way calling
service. We add the call waiting feature and try to see if some of the actions of this
new feature risk nondeterminism with respect to existing features.

The method is applied  in two steps. First we must prove that a backward
path satisfying all conditions of synchronizing processes, starting with the action

||

|||

|||

|||

[]

gathering traceverifying trace

Fig.3 Backward reasoning on a parse tree like representation

flash_hook, does exist within the first feature, in our case the three-way calling.
This path is summarized in Fig. 4. Then, given the behavior expression after pro-
ducing this path, we attempt to execute a second path starting from the call wait-
ing flash_hook action. The resulting path is summarized in Fig. 5. Since the two
paths are executable within each feature, we can conclude that it is possible for
the flash_hook action to present itself in a situation of nondeterminism.

We now describe how the first path is derived. Processes involved in this path
have been shaded in Fig. 4. Boxes represent behavior expressions or instantiated
processes. Let’s start with the flash_hook. Step 1 tells us to look for more occur-
rences of a flash_hook action. We find out that there is one in the three-way call-
ing feature. Step 2 tells us to verify if the behavior expressions where the two
flash_hook actions belong are in a situation of interleaving. This is indeed the
case, as it can be seen by simple inspection. Having these two behavior expres-
sions, which happen also to be processes, we can start to analyse how they can be
executed in interleaving.

We therefore start to execute the three-way-call feature backward. First we
observe that our two features are instantiated in a process called features that
itself is instantiated in a process completed_connection:

processcomplete_connection[u,n]  (NUMBER:phone_number,
                                                             SUBS_SERV:subscrib_services):noexit:=
               n ! NUMBER ! connect
           ;
              (
                   user_events[u,n](NUMBER,SUBS_SERV)
                 |||
                   features[u,n](NUMBER,SUBS_SERV)
             )
endproc

Within the above process we discover that our instance of process features is
further interleaved with user_events. Inspection of this second process leads to
the conclusion that there is nothing of interest for our problem in it, consequently
we keep going backward following action prefixes to end up with the n ! NUM-
BER ! connect action. This action is important because it tells us that a full con-
nection exists when the call_waiting feature is activated.

We are now at the top of the behavior expression of process complete_connec-
tion. We now must look for an instance of this process. We find one of interest in
process call_initiator_phone. We observe that we can follow a series of action pre-
fixes all the way back to the top of the behavior expression of this process to the
off_hook action. Again we look for instances of this process and we find that the
only case is in process phone. Thus we know the path that leads to a full connec-
tion of a phone.

Fig. 4   Call waiting interaction with tree way calling summary

Fig.  5  Call_waiting interaction with tree way calling summary

Process phone has many instances that are all in parallel with an instance of pro-
cess network. First we must pick one of these instances. This has the effect of
instantiating the value of the formal parameter number. In our case we have
picked the instance of phone[u,n](num1). Concerning this choice, as well as the
other choices of parameters that are necessary at this level, note that since the
method consists of finding at least one scenario that verifies our goal, any appro-
priate values can be supplied by the user. In simple phone specifications these
values are usually phone numbers that decide which phones are involved in com-
munication. Here we only need to establish a connection between a phone that
has both features (call waiting and three way calling) and a sort of neutral phone
(without any features) to avoid interferences from features that are not presently
of interest. The instance of process phone[u,n] that we have selected belongs to a
behavior expression that is in parallel with an instance of process network. We
must now verify if the collected trace satisfies the conditions of process network.

Thus we have obtained, by backward execution, the following instantiated
trace, which we call T3waycall :

u ! num1! off_hook -->
     n ! num1! tone -->
          n ! num1! conreq ? num2 ( instance of variable Called_num) -->
              n ! num1! connect -->
                    u ! num1! flash_hook --> ...

 In other words, the trace must be composed with the remaining high-level
behavior expression:

             ( T3waycall ||| phone[u,n](num2) ||| ... ||| phone[u,n](num5) )
        |[n]|
             network[n]

Thus we must try to compose T3waycall  with the network, this time in forward
execution, keeping in mind that any actions of process network that cannot syn-
chronize with the trace  could synchronize with the remaining behavior expres-
sions that are interleaved with the trace (here phone[u,n](num2) ||| ... |||
phone[u,n](num5) ).

One available technique to help in this process, is to apply goal oriented exe-
cution taking the sequence of actions of trace T3waycall  as a goal. However we shall
continue describing the process as if it were purely manual one, in order to
explain the steps.

Thus, let us examine process network.

processnetwork[n,a]:noexit:=
         n ? CALLER:phone_number ! tone
      ;
      (
            connection_handler[n,a](CALLER)
          |||

            network[n,a]
      )
endproc

The network’s first action is to provide a dial tone. This matches with the sec-
ond action of T3waycall.

 The execution of this action instantiates also the variable CALLER to the
value num1 by value passing.

The next step is to instantiate process connection_handler, noting that we also
have the possibility to interleave recursively with process network.

processconnection_handler[n,a](CALLER:phone_number):noexit:=
             n ! CALLER ? CALLED:phone_number ! conreq ?  SCR_LIST:screen_list
         ;
         (
            [ CALLED NotIn SCR_LIST ] ->
                      n ! CALLED ! ring
                   ;
                      (
                           n ! CALLED ! connect
                         ; n ! CALLER ! connect
                         ; relay_user_events[n](CALLER,CALLED)
                       []
                         ...
endproc

Again we see that the next action in T3waycall can match the first action of pro-
cess connection_handler, i.e. n ! CALLER ? CALLED:phone_number ! conreq , but
then the next action from T3waycall ( n ! NUMBER ! connect ) can no longer match
the remaining sequence of process connection_handler. Thus at this point we
have to go back to the many instances of process phone that are interleaved with
T3waycall to see if one of these instances can provide a matching action to the ring.
We pick instance phone[u,n](num5). Eventually we reach the connect action that
matches our call initiator. This was the last action before the flash_hook action.
No other actions from the network process are required in order to proceed in
T3waycall. So far this exercise has produced a full connection between phones 1 and 2
that is necessary in order to reach the flash_hook action of feature three-way call.

Now we are back to the original problem: how can process call_waiting be acti-
vated when a full connection is established ? This is the second path to explore as
summarized in Fig. 5. Processes involved in this path have been shaded. As we
mentioned at the beginning of this exercise, the second path will be derived with
all behavior expressions starting in the state they have reached while deriving
the first path. We can use again the same kind of backward reasoning, leaving
the details to the reader.

We start with the following behavior expression that results from the back-
ward execution of the first feature (where Rest of3way_call is now reduced to the
flash_hook action and what follows it).



15 March 1, 1995 6:15 pm

      ( ( Rest of3way_call ||| Call_waiting(num1)) ||| talking[u](num2)  |||  ... ||| phone[u,n](num5)   )
   |[n]|
      ( Stop ofconnect_handler(num1,num2)||| network[n] )

The call waiting trace will have to find a matching trace in the interleaved
process network and there we can expect that some synchronization will be
required with one of the remaining interleaved phone instances.

When this is done, we have proof that nondeterminism can exist on the flash_-
hook trigger for the three-way-calling and the call-waiting features.

The same reasoning can be applied to the call waiting, call forward on busy
example. This time it is the ring action that is ambiguous.

2.6 A tool
The backward execution method used in step 3 has been implemented in a

prototype tool programmed in Prolog. The backtracking feature of Prolog is
uniquely suited for this purpose. The internal representation of LOTOS specifica-
tions in the tool is quite unlike what is found in usual interpreters or expanders.
We have used an inverted data base which makes it possible to walk a behavior
both top- down and bottom-up. Actions, process identifiers, etc., are uniquely
labelled so that the user can instruct the tool to direct execution to those she
wants to use. No new inference rules are needed, the only novelty being in the
way the inference rules are applied.Our tool does not try to validate parameter
values with relation to expressions contained in parameter lists.  For example, if
we have an instantiation such as P[...](...,n-1,...) for a process declaration
P[...](...,m,...) it would be necessary to see that n has value m+1.  Currently, it is
responsibility of the user to choose values so that such consistency is assured.  In
a more sophisticated tool, values would have to be derived automatically when-
ever possible, or at least consistency checks would have to be included.

To carry out easily the type of analysis advocated in this paper, it would be
desirable to include a backward execution facility in conventional LOTOS inter-
preters, together with features to replace behavior expressions with the obtained
traces, and then continue with forward execution (possibly goal-oriented), as we
have shown.

3. CONCLUSIONS

Detection of feature interaction is an existence proof. We explore all the poten-
tial alternatives until we find a symptom of  interaction. Backward and forward
execution work together to reduce the number of cases to be considered. The
backward trace from the point of interest acts like a partial test case where a
number of interleaved actions are missing and can be filled by choices of the user.

We have shown that this technique can be facilitated by the use of appropriate
tools.  The use of appropriate specification styles may also be important, although
we have insufficient experience in this respect.

This technique has other applications beyond the one discussed in this paper.-
For example, it allows generating meaningful scenarios or use cases, starting

from the desired goals, in a stepwise fashion and with meaningful values.

Annex: The specification

(* ----------
Note1: Termination sequences have been left out of the specification because they are irrelevant
for the purpose of the paper

Note 2: We have used the LOTOSPHERE enhanced data types, as described in [Sto92]
------------*)

specification feature_interaction_system[u, n, a]:noexit
(* written by Bernard Stepien, November 1993 *)

   library Boolean, NaturalNumber , Set endlib

 enumtype phone_number is
   { num1, num2, num3, num4, num5 }
 endtype

 enumtype primitives is
   {
     offhook, conreq, ring, answer, connect, talk, pwd, dial, tone,
     star, pound, voice_mail_answer, busy, flash_hook,
     conreq_calling_card, hang_up, discon_req,
     activate_call_forward,
     play_announce_pwd, deliver_messages,
     play_announce_star_pound, play_announce_management,
     play_announce_new_number, good_bye, detect_forward,
     unconditional_refusal
   }
    subclass

      user_actions  { offhook, answer, talk, pwd, dial, star, pound, flash_hook,
                       hang_up, activate_call_forward }
      network_actions  { tone, conreq, ring, connect, busy, detect_forward, unconditional_refusal }
      voice_mail_actions  { voice_mail_answer, play_announce_pwd,
                             deliver_messages, play_announce_star_pound,
                             play_announce_management,
                             play_announce_new_number,
                             good_bye
                           }
 endtype

 enumtype service is
   {
     three_way_calling, call_waiting, call_forward, originate_screening }
 endtype

 enumtype active_service is
   { no_active_service, forward_calls }
 endtype

 settype subscrib_services is service
    elements service

    values
      no_services = { }
 endtype

 settype screen_list is phone_number
   elements phone_number
   values
    no_screen_list = { }
 endtype

 behavior

 (
      phone[u, n](num1, { three_way_calling, call_waiting,originate_screening }, no_active_service , num1, {
num4 } )
    |||
     phone[u, n](num2, { call_forward }, no_active_service ,num4, no_screen_list)
    |||
     voice_mail[u, n](num3)
    |||
     phone[u, n](num4,no_services, no_active_service,num4, no_screen_list)
    |||
     phone[u, n](num5,no_services, no_active_service,num5, no_screen_list)
 )

|[n]|

 network[n,a]

 where

 processphone[u, n](NUMBER:phone_number,SUBS_SERV:sub-
scrib_services,ACTIV_SERV:active_service,FWD_NUMBER:phone_number, SCR_LIST:screen_list):noexit:=

    call_initiator_phone[u, n](NUMBER,SUBS_SERV,ACTIV_SERV,FWD_NUMBER, SCR_LIST)

    []

    call_responder_phone[u, n](NUMBER,SUBS_SERV,ACTIV_SERV,FWD_NUMBER, SCR_LIST)

 endproc

 processcall_initiator_phone[u, n]
         (NUMBER:phone_number,SUBS_SERV:subscrib_services,ACTIV_SERV:active_service,
                                                  FWD_NUMBER:phone_number, SCR_LIST:screen_list):noexit:=

   u ! NUMBER ! offhook
   ; n ! NUMBER ! tone
   ;
   (
   (
          n ! NUMBER ? CALLED_NUMBER:phone_number ! conreq ! SCR_LIST
     ;

        (
          complete_connection[u,n](NUMBER,SUBS_SERV)
          []
          n ! NUMBER ! CALLED_NUMBER ! unconditional_refusal
          ; phone[u,n](NUMBER,SUBS_SERV,forward_calls,FWD_NUMBER,SCR_LIST)

      )
    []
          calling_card_call[u,n](NUMBER, SUBS_SERV)
   )
   []
    (
       u ! NUMBER ! activate_call_forward ? FWD_NUMBER: phone_number
     ; phone[u,n](NUMBER,SUBS_SERV,forward_calls,FWD_NUMBER,SCR_LIST)
    )
    |||
    busy_ring[u,n](NUMBER,SUBS_SERV)
   )

  endproc

   processcall_responder_phone[u, n]
         (NUMBER:phone_number,SUBS_SERV:sub-
scrib_services,ACTIV_SERV:active_service,FWD_NUMBER:phone_number, SCR_LIST:screen_list):noexit:=

   n ! NUMBER ! ring
   ;
   (
     (
       (
         [ ACTIV_SERV ne forward_calls ]->
              u ! NUMBER ! answer
         ;    n ! NUMBER ! connect
         ;    u ! NUMBER ! talk
         ;    stop
       )

      []
       call_forward[u,n](NUMBER,SUBS_SERV,FWD_NUMBER,SCR_LIST)
     )
    |||
     busy_ring[u,n](NUMBER,SUBS_SERV)
   )

 endproc

 processuser_events[u,n](NUMBER:phone_number, SUBS_SERV:subscrib_services):noexit:=

       u ! NUMBER ! talk
     ; user_events[u,n](NUMBER, SUBS_SERV)

    []

       u ! NUMBER ! star
     ; n ! NUMBER ! star

     ; user_events[u,n](NUMBER, SUBS_SERV)

    []

       u ! NUMBER ! pound
     ; n ! NUMBER ! pound
     ; user_events[u,n](NUMBER, SUBS_SERV)

    []

       u ! NUMBER ! pwd
     ; n ! NUMBER ! pwd
     ; user_events[u,n](NUMBER, SUBS_SERV)

    []

      (* after the network played a dial a new number announcement *)
       n ! NUMBER ? NEW_NUMBER:phone_number ! conreq
     ; complete_connection[u,n](NUMBER, SUBS_SERV)

    []

       u ! NUMBER ! hang_up
     ; n ! NUMBER ! discon_req
     ; stop

 endproc

 processcalling_card_call[u,n](NUMBER:phone_number, SUBS_SERV:subscrib_services):noexit:=

     n  ! NUMBER ? CALLED_NUMBER:phone_number ! conreq_calling_card ? CHARGE_NUMBER:-
phone_number
   ; complete_connection[u,n](NUMBER, SUBS_SERV)

 endproc

  processcomplete_connection[u,n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=

       n ! NUMBER ! connect
     ;
     (

   user_events[u,n](NUMBER,SUBS_SERV)
|||
   features[u,n](NUMBER,SUBS_SERV)

     )

  endproc

  processfeatures[u, n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=

      call_waiting[u,n](NUMBER,SUBS_SERV)

     |||

      three_way_calling[u,n](NUMBER,SUBS_SERV)

  endproc

 processcall_waiting[u,n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=

    [ call_waiting IsIn SUBS_SERV] ->
         n ! NUMBER ! ring
       ; u ! NUMBER ! flash_hook
       ; n ! NUMBER ! connect
       ;
        ( user_events[u,n](NUMBER,SUBS_SERV)
          [>
            (   n ! discon_req

      ; features[u,n](NUMBER,SUBS_SERV)
            )

)

  endproc

  processthree_way_calling[u,n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=

    [ three_way_calling IsIn SUBS_SERV] ->
        u ! NUMBER ! flash_hook
      ; u ! NUMBER ! dial ? C:phone_number
      ; n ! NUMBER ! conreq ! C
      ; n ! NUMBER ! connect
      ;

( user_events[u,n](NUMBER,SUBS_SERV)
          [>
            (   n ! discon_req

      ; features[u,n](NUMBER,SUBS_SERV)
            )

)

  endproc

  processcall_forward [u,n](NUMBER:phone_num-
ber,SUBS_SERV:subscrib_services,FWD_NUMBER:phone_number,SCR_LIST:screen_list):noexit:=

    [ call_forward IsIn SUBS_SERV] ->
 n ! NUMBER ! detect_forward ! FWD_NUMBER

       ; phone[u,n](NUMBER,SUBS_SERV,forward_calls,FWD_NUMBER,SCR_LIST)

  endproc

  processbusy_ring[u,n](NUMBER:phone_number,SUBS_SERV:subscrib_services):noexit:=

    [ call_waiting NotIn SUBS_SERV] ->
 n ! NUMBER ! busy

       ; u ! NUMBER ! hang_up
       ; n ! NUMBER ! discon_req
       ; features[u,n](NUMBER,SUBS_SERV)

  endproc

  processvoice_mail[v, n](NUMBER:phone_number):noexit:=

        n ! NUMBER ! ring
   ;    v ! NUMBER ! voice_mail_answer
   ;    n ! NUMBER ! connect
   ;    v ! NUMBER ! play_announce_pwd
   ;    n ! NUMBER ! pwd
   ;    v ! NUMBER ! deliver_messages
   ;    v ! NUMBER ! play_announce_star_pound
   ;
   (
        n ! NUMBER ! star
     ;  v ! Number ! good_bye
     ;  stop
    []
        n ! NUMBER ! pound
     ;  v ! NUMBER ! play_announce_management
     ;  stop
   )

 endproc

 processnetwork[n,a]:noexit:=

   n ? CALLER:phone_number ! tone
   ;
   (
     connection_handler[n,a](CALLER)
     |||
     network[n,a]
   )

 endproc

 processconnection_handler[n,a](CALLER:phone_number):noexit:=

   n ! CALLER ? CALLED:phone_number ! conreq ? SCR_LIST:screen_list
   ;
   (
     [ CALLED NotIn SCR_LIST ] ->
     n ! CALLED ! ring
     ;
     (
         n ! CALLED ! connect
       ; n ! CALLER ! connect
       ; relay_user_events[n](CALLER,CALLED)

      []

 n ! CALLED ! detect_forward ? FWD_NUMBER: phone_number
       ; n ! FWD_NUMBER ! ring
       ; n ! FWD_NUMBER ! connect

       ; n ! CALLER ! connect
       ; relay_user_events[n](CALLER,FWD_NUMBER)
     )

 []

   [ CALLED IsIn SCR_LIST ] ->
      n ! CALLER ! CALLED ! unconditional_refusal
      ; stop
   []
      n ! CALLED ! busy
      ; stop
   )

 []

   n ! CALLER ? CALLED_NUMBER:phone_number ! conreq_calling_card ? CHARGE_NUMBER:phone_-
number
   ; network_complete_connection[n,a](CALLER, CALLED_NUMBER)

 endproc

 processrelay_user_events[n](CALLER,CALLED:phone_number):noexit:=

   n ! CALLER ? EVENT:primitives [ is_user_actions(EVENT) ]
   ; n ! CALLED ! EVENT
   ; relay_user_events[n](CALLER,CALLED)

  []

   n ! CALLER ! discon_req ; stop

 endproc

  processnetwork_complete_connection[n,a](CALLER,CALLED:phone_number):noexit:=

          n ! CALLED ! ring
     ;    n ! CALLED ! connect
     ;    n ! CALLER ! connect
     ;
     (
            relay_user_events[n](CALLER,CALLED)
       |||
            calling_card_intercept[n,a](CALLER)
     )

  endproc

  processcalling_card_intercept[n,a](CALLER:phone_number):noexit:=

           n ! CALLER ! pound
        ;  a ! CALLER ! play_announce_new_number
       ;  n ! CALLER ? NEW_NUMBER:phone_number ! conreq

        ;  network_complete_connection[n,a](CALLER,NEW_NUMBER)

  endproc
 endspec


